
The Best Kept Secret of Custom Tasks Built in SAS® Studio! 

ABSTRACT  
SAS Studio is a web-based tool that allows users to run SAS code through the web browser. It also offers 
several predefined tasks, which allow for point-and-click user interfaces for several analytical procedures. 
A big surprise to many SAS programmers is that SAS Studio offers the capability to create custom tasks 
with very little effort. These custom tasks can become powerful tools of automation for many audited 
processes. Custom tasks can be created to provide a user interface requiring entry of a few parameters 
that can set off a series of SAS programs fully automated to run, review log files for errors, check data 
quality, send emails, and publish reports. Custom tasks to automate SAS programs can be created to 
meet the most stringent data control requirements, including SOX (Sarbanes-Oxley), MRM (Model Risk 
Management) and FRB (Federal Reserve Board) audit reviews imitating a production environment for 
publication of financial reports such as CECL (Current Expected Credit Losses), DFAST (Dodd-Frank Act 
Stress Test), and CCAR (Comprehensive Capital Analysis and Review). With the addition of some well-
known Base SAS features to read directories, capture logs and send emails, a custom task can set off a 
fully functional and controlled production environment with less effort than most imagined.  

INTRODUCTION  
This paper assumes the reader has an intermediate to advanced level of SAS programming with Base 
SAS, and SAS macro language. To demonstrate how to use a custom task to create a fully functional 
production process several different uses of Base SAS and SAS macro language will be demonstrated. 
First, we will introduce a provided SAS Studio task and then show how to build a simple custom task that 
executes SAS programs. Second, we will show some SAS macro language that can be added to build 
directories, copy SAS programs, capture logs, read logs and provide artifacts required for audit purposes. 
Finally, we will identify some additional techniques for the custom task to send emails. There are many 
papers and resources to help build complex custom tasks in SAS Studio, but the purpose of this paper is 
to show how a custom task with a few additional techniques can represent an automated production 
process with controls so it can run without any changes and capture all artifacts required by several 
internal and external audit reviews. The programs shown in this paper are using SAS Studio Enterprise 
Edition Release 3.6, SAS release 9.04 on a Windows platform.  

SAS STUDIO TASKS AND CUSTOM TASKS 

SAS STUDIO TASKS 
SAS Studio can be used to write and execute Base SAS code in the Program Editor; but it also provides 
built-in tasks that allow a user to easily run some simple analytics with a few clicks. 

The point-and-click user interfaces offered by SAS Studio can guide a user to analyze, report, graph or 
convert data without knowing Base SAS. Some of the tasks that come with SAS Studio include the 
following:  

List Table Attributes Task – utilizes PROC DATASETS (Data Task) 

Characterize Data – utilizes PROC FREQ (Data Task) 

List Data – utilizes PROC PRINT (Data Task) 

Bar Chart – utilizes PROC SGPLOT (Graph Task) 

Summary Statistics – utilizes PROC MEANS (Statistics Task) 

The tasks are all very helpful and since they show the SAS code related to each task, they can be used to 
help beginner SAS programmers learn SAS. 

The screenshot below shows a SAS Studio screen upon login. Notice the Task selections listed in the left 
navigation pane. If you do not see Tasks, check your View options using the More Applications icon 
denoted by small circle with lines to the right of SAS Programmer.  

 
The screenshot above has the Data tasks expanded to show you a list of the predefined tasks provided 
by SAS Studio to run some simple analytics related to SAS data.  

The screenshot below shows a split task pane after the List Table Attributes task is selected. The Settings 
pane (left side) has three tabs called DATA, OPTIONS and INFORMATION. The Code/Results pane 
(right side) has three tabs called CODE, LOG and RESULTS; similar to a Program Editor pane.  

 
 

The screenshot above is showing two tabs of the tasks: the DATA tab which displays the user interface 
and the CODE tab showing the SAS code associated with this task. The SAS code for this task is a 
PROC DATASETS procedure. Once the user selects a dataset, the task can be executed by selecting the 
running man. This will produce the results of the SAS code execution.  

If the user selects to view the OPTIONS tab in the Settings, the screenshot below is presented. It 
provides different options for the user to select when the task is executed. As a default the task will 
produce the Data set attributes, and a Variables list.  

 
 

The INFORMATION tab shows some information relevant to the task that the user might find helpful to 
understand the code executed as part of this task.  

 
 

The screenshot below shows the RESULTS after a SAS dataset is selected and the task runs.  

The LOG tab shows the SAS code that ran. If errors or warnings had occurred it would be denoted in the 
Errors, Warnings and Notes section. 

 
 

So far, the CODE ran for this task, the LOG and the RESULTS should all look familiar to any SAS 
programmer. A SAS programmer could produce the same results by running a SAS program with the 
following code:  
ods noproctitle; 

ods select attributes variables; 

proc datasets; 

  contents data=PRODDATA.NATIONAL_MEV_31MAR2020 order=collate; 

quit; 

If the program needed to be ran by multiple users or several times a SAS programmer could also modify 
the code to use macro language as shown here:  
%let dataname = PRODDATA.NATIONAL_MEV_31MAR2020; 

ods noproctitle; 

ods select attributes variables; 

proc datasets; 

  contents data=&dataname. order=collate; 

quit; 

Another use of macro language to produce the same results could also be written as shown below: 
ods noproctitle; 

ods select attributes variables; 

%macro list_attributes_table(dataname); 

proc datasets; 

  contents data=&dataname. order=collate; 

quit; 

%mend list_attributes_table; 

%list_attributes_table(PRODDATA.NATIONAL_MEV_31MAR2020); 

Both sets of SAS code above would produce the same results as the task. The biggest and most 
important difference is to select a different dataset the SAS program would need to be changed.  

When a set of SAS programs need to run daily, weekly, monthly, or quarterly to support a regulatory 
process that requires full automation from the data source to the published results-- this simple change to 
the SAS program is considered a ‘change’ that might require full production change control processing.  

This is one of the biggest advantages of a custom task; it will allow a consistent process that executes 
SAS code to run without changing any SAS programs related to the process. Any changing information 
required by the SAS programs such as dates, output locations, etc. can be entered into the process 
through a custom task user interface. The parameters entered in the user interface get passed to the SAS 
code similar to macro variables but without actually touching the SAS program.  

Now that you’re more familiar with SAS Studio tasks, let’s look at producing a custom task to execute a 
set of SAS programs.  

CUSTOM TASKS 
For the purposes of this paper, I will show how I learned to build a custom task that automated a monthly 
process without ever changing or touching any SAS programs (as they were considered in production 
and would require change management control processing – a time consuming IT defined procedure only 
meant to be used when actual material logic changes were being made, not to simply change input 
parameters). 

There are a few ways to create a custom task, but first I wanted to see what was happening behind the 
scenes of the tasks. 

I chose to click the New Task icon and select Sample Task as the screenshot shows below.   

 
Once selected you will see color coded logic that builds the Sample Task. 

All tasks are based on the CTM (Common Task Model), which is made up of XML (Extensible Markup 
Language – similar to HTML, but without predefined tags to use), and VTL (Velocity Template Language). 

I had some experience with JavaScript and HTML tags, so this logic looked familiar, but I certainly did not 
want to build anything from scratch. I’m a SAS programmer, not a web developer!  

Instead, I ran this Sample Task by clicking the running man to see what user interface options it was 
demonstrating that I could use for my custom task. The logic in the Sample Task and Advanced Task are 
good resources to find examples you need if you want to build a more complex custom task.   

The screenshot below shows the results of running the Sample Task, and I selected the OPTIONS tab to 
view the different options available. I expanded the options that most interested me.  

 
The CHECK BOX, and DATE PICKER were options I thought I could use for my custom task.  

Next, I went back to the tab that showed all the language behind these elements and selected the code I 
would need for my own custom task. By searching the language in the Sample Task, I found the code 
needed for these options.  
Note: there are several very good conference materials available with information about all the options and how to 
build a custom task from scratch; these will be listed in the references. 

It was easy to identify the logic for the elements that interested me. 

CHECK BOX: 

 
DATE PICKER: 

 
Now I started my custom task, by selecting and opening the “New Task” template as shown by the 
screenshot below. 



7 

 
As mentioned earlier the custom task is built using HTML. The HTML code that the custom task uses has 
a framework and the framework includes task elements such as REGISTRATION, METADATA, UI, 
DEPENDENCIES, and a CODE TEMPLATE. Note: the new task does not show DEPENDENCIES element and I 
have never used this element. It will not be discussed in this paper. 

We will modify the new task template and update all the task elements shown in the screenshot below as 
needed for our custom task. 

 
The language behind SAS Studio tasks is a language of its own, developed outside of SAS. The 
“Registration” is not registering your task, it is simply a header that documents the task. The ‘Metadata’ is 
information for the task. The ‘Options’ are more like tags to identify anything used in the task.  

There are some coding items to note. Comments begin with # or <--. Variable names assigned are 
prefixed with $. The color coding in the editor can help identify errors. Full documentation with more 
details about creating custom tasks can be found in SAS Studio: Developer’s Guide to Writing Custom 
Tasks available from SAS Customer Support Site | SAS Support. 

REGISTRATION ELEMENT 
The Registration element provides information about the task and is very simple to update. It contains 
some child elements: Name, Description, GUID, Procedures, Version and Links. I updated the Name and 
Description, but leave the values for GUID, Procedures, Version and Links as they are generated; see 
screenshot below. 

https://support.sas.com/en/support-home.html

Once I made my changes to the Name and Description, I save the task as shown in the screenshot 
below. I save the task to a folder designation (instead of My Tasks) so that other users with access can 
also run the custom task. (At our company, the custom task is saved and tested on a test SAS server and 
then it is moved by a deployment team to a secured folder on the production SAS server that only limited 
production users can access.) 

 

METADATA ELEMENT 
The Metadata element contains child elements of DataSources and Options.  

The DataSources element contains a child element DataSource. The Datasource element specifies 
information about the dataset and is required if you choose to have a DATA tab in your task that requires 
the user to select a SAS dataset.  

I have never built a custom task where I have needed the user to select a SAS dataset; therefore, I have 
never used this child element and do not require a DATA tab in my custom task. If you recall, the 
screenshot of the List Table Attribute task had DATA, OPTIONS and INFORMATION tabs in the Settings 
panel. (See screenshot below)  

 
Note: if you are interested in building a custom task with a DATA tab there are several very good conference 
materials available; these will be listed in the references. 

Since I do not need a DATA tab, I deleted the DataSources tags (<DataSources></DataSources>) and I 
only need to update the Metadata child element called Options.  

The Options element contains a child element Option. Option is the only child element of Options but is 
required for the task to run.  

I identified the Option element for the Options Tab and copy Option element information for it from the 
Sample Task code and paste it in my custom task within the Options tags (<Options></Options>). 

 
Now I paste in the Option information I saved earlier from the Sample Task about CHECK BOX and 
DATE PICKER.   

 
I’ve completed with the changes I wanted to make to the Metadata element. The options are required to 
run the task, but if I run my custom task at this point nothing will show since I have yet to enter 
information in the UI element, see the screenshot below. 

 

UI ELEMENT 
The UI element contains child element of Container. Container contains child element of Group. The 
Group element contains child elements of DataItem, OptionChoice and RoleItem. The OptionChoice 
contains OptionItem as a child.  

For my purposes I go back to the Sample Task code to identify and copy what I need from the UI element 
based on the options I am using. I need the Container, Group and OptionItem elements for my Options 
Tab, and the Group option with OptionItem child elements for the CHECK BOX and DATE PICKER. 
Finally, the closing Container tag. 

Once these are entered in the custom task between the UI tags, I can test the basic functionality of my 
custom task by clicking the running man. There are few types of errors you might get when trying to save 
the custom task, the following is one example.  

 
If this happens take a close look at your elements and make sure there is an open and close tag for each 
element. Also make sure all the elements named in the Metadata-Options section match up with the 
elements named in the UI section. 

The image below shows how all tags must be balanced. It also shows how the names in the Metadata-
Options element section must have matching names in the UI element section. 

 
If you can save your new task without an error, you can test your task by clicking the running man to run 
the Run_Monthly_Report.ctm, (Note: we have not yet updated the CODE TEMPLATE element but the new task 
came with a proc print statement in this element). 

Running the Run_Monthly_Report.ctm will present the following as shown in this screenshot. 

Notice the CHECK BOX and DATE PICKER have arrows to the left that need to be selected to expand 
these selection options. The screenshot below shows what this looks like after both options are 
expanded. 

 
At this point, we successfully built a user interface with two selection options, but we have not added the 
actual SAS code we want the custom task to execute. Also, we have not incorporated our selection 
options, so they pass into our SAS code as macro variables. This is all done in the CODE TEMPLATE 
element. 

Before we move forward to add the SAS code that we want our custom task to execute, it’s a good idea 
to go into your custom task and modify some names and labels so they are meaningful for your users and 
process. 

After making the appropriate changes the custom task looks like the screenshot below. 

 
I have removed the extra Group option at the top of the OPTIONS tab as I don’t need it. I’ve also added 
the open=”true” to the CHECK BOX and DATE PICKER options so the user does not have to expand 
them. You will also note I modified some text to help instruct the user how to use the CHECK BOX and 
the DATE PICKER. I’ve also changed the name of my options to: chkData and dateMonthEnd. 

Now if I run Run_Monthly_Report.ctm the user interface looks like the screenshot below. 

The new custom task looks like it is ready to run SAS code. Note that the InputType=”string” will not 
generate variable values that can be used in the SAS code. Only the CHECK BOX and the DATE 
PICKER will create variables that can pass into any SAS code within the CODE TEMPLATE element, as 
these were the only selection options we added to our custom task. 

CODE TEMPLATE ELEMENT 
The new task came with some simple SAS code contained in the CodeTemplate element.  

 
We are going to replace this code with some SAS macro language and SAS code to have the custom 
task execute our monthly report. 

First, I like to set all the variables created by the selection options to macro variables. You could 
reference variables created in a task within SAS code with $chkData or $dateMonthEnd so it is not 
necessary but since I’m more familiar with SAS macro language I do it.  

Also, note all text within the CodeTemplate tags can be normal SAS code, so comments can be made 
with /* */ or * ;.  

Here is the SAS code I will first add to set my variables to macro variables.  
/* Create macro variable chkData*/ 
%let chkData = $chkData; 
/* Create macro variable dateMonthEnd */ 
%let dateMonthEnd = $dateMonthEnd; 
/* Show values of the macro variables */ 
%put &chkData.; 
%put &dateMonthend.; 
 

I like to comment thoroughly so any users understand what SAS code the custom task will run and what it 
is doing. Also, I make use of the %put statement to ensure what the values of the macros are during the 
custom task execution.  

Let’s take a look at the custom task after this code is added to the CodeTemplate element. 

Now if I run Run_Monthly_Report.ctm with this change the user interfaces looks like the screenshot 
below. 

 
As the user makes selections from the options you will see the values change on the CODE tab in the 
Code/Results panel on the right. Take a look at the screenshot below. 

 
Notice in the %let chkData statement: chkData = 1 because the CHECK BOX is selected. Also, the user 
selected a date and the %let dateMonthEnd statement now shows: dateMonthEnd = Jul2022, but this is 
not the date value I want in my macro variable &dateMonthEnd. In order to change it I have to make a 
change to the option statement in the Metadata-Options element section. 

It has a format of monyy7. but I want to change this to a format of Date9. because I prefer to work with 
this date format in my SAS code, so it becomes: 

 
Now that I’ve made that change the macro variable value show in a Date9. Format. 

We want to be sure the date selected is a month end date since the data the SAS process will read is 
created with snapshots of all records on the last day of each month. If the date entered is not a month 
end date the process would not select any data records so it’s not worth running. The &dateMonthEnd 
macro variable value can be checked with the following SAS macro language. 

/* Check the dateMonthEnd is actually the last day of the month */ 
%macro check_date_month_end; /* Create new macro variable: end_month_dt */ 
data _null_; 
  format end_month_dt best1.; 
  if intnx('month', "&dateMonthEnd."d, 0, 'e')= "&dateMonthEnd."d then  
  end_month_dt=1; 
    else end_month_dt=0; 
  /* Create macro variable end_month_dt */ 
  call symput('end_month_dt',end_month_dt);  
run; 
%mend check_date_month_end; 
%check_date_month_end; 
%put &end_month_dt.; 
 

Once the logic above is added I can use it to control if the custom task runs the SAS program that 
publishes the monthly report with another macro shown here: 

%macro run_monthly_report;  
 %if &chkData. eq 1 %then %do;  

%if &end_month_dt. eq 1 %then %do; 
      %include "&prodpath.\&mepath.\SAS_Programs\Run_Monthly_Report.sas"; 
   %end; 
   %else %if &end_month_dt. eq 0 %then %do; 
    DATA error_msg; 
     format var $45.; 
     var= 'ERROR: MUST BE LAST DAY OF THE MONTH'; 
    run; 
    Title "Monthly Run for &dateMonthEnd. Was aborted"; 
    proc print;  
    run; 
    title;   
   %end; 
 %end; 
 %mend run_monthly_report; 
%run_monthly_report; 
 
 

Some of you may wonder about the &chkData. macro variable created from the CHECK BOX option. 
There are many scenarios where this macro variable can be useful in controlling SAS to run other steps. 
In the scenario so far, we expect the box to be checked for the SAS program designated to be called and 
the monthly report process to run. At this point if the user does not check the box nothing would run. You 
will see more about this macro variable later in the paper. 

After these changes we need to test the logic added. This time the user will select a date that is not a 
month end date and see if SAS produces an Error message in the RESULTS tab.  

Notice the %let dateMonthEnd = 01Aug2022 in the screenshot above. This will test our new logic.  

When we click the Running Man while in a task, SAS Studio submits the SAS code added to the 
CodeTemplate element, and we will have logs and results to review when it is done. We have not done 
this yet since we didn’t have any real SAS code in our CodeTemplate for SAS to execute.  

 
SUCCESS! The SAS macro language worked and identified the date selected is not the last day of the 
month and generated the Error Message. 

Here is a copy of the log from our custom task, Run_Monthly_Report.ctm 

The LOG shows the values of the macro variables created because I used %put statements. I’ve 
highlighted this with the boxes. I find this to be very useful when troubleshooting any issues.  

FINAL CUSTOM TASK 
All elements of the custom task are complete. The SAS code contained in the Run_Monthly_Report.sas 
is still a mystery. At this point in the paper it will have the following SAS code to produce a simple report 
from data read in from SQL. 

Finally, we are ready to run our custom task, Run_Monthly_Report.ctm. It will pass the macro variable 
&dateMonthend, so it can be used in the SAS program called with the %include statement. We’ve made 
sure if the date selected is not a month end date nothing will run, and we are ready to run the report.  

The final results of running the custom task with the $chkData = 1, and $dateMonthEnd = 30Jun2022 are 
shown in the screenshot below. 

 
This first section only did some of the work required for automation of the monthly process. It allowed the 
monthly process to run without any changes being made to the main SAS program – 
Run_Monthly_Report.sas. It also does not require any changes to the custom task created called 
Run_Monthly_Report.ctm in order for it to be used monthly.  

However, at this point the user had to assume the macro variables &prodpath. and &mepath. used to 
define a libname and store the report within the Run_Monthly_Report.sas were already created. The user 
also had to visually review the final log for any errors, or warnings after the custom task ran. Most audit 
reviews require to see the SAS log file showing the process ran without errors so the user would have to 
manually save the information on the LOG tab after the process ran to store it.  

This next section will show some techniques that have come up in audit reviews and will further automate 
the monthly process.  

AUTOMATION TECHNIQUES – NEW DIRECTORIES, VERSION CONTROL, LOG 
STORAGE AND CHECKING 
Before we move into the automation techniques, I want to explain how changes are being made to the 
custom task, Run_Monthly_Report.ctm 

Since I saved the Run_Monthly_Report.ctm to a folder called SAS_Tasks\ as shown above, I can double 
click it from the left Servers Files and Folders pane when in SAS Studio. When I do this SAS Studio will 
ask if I want to edit or run the task. If I want to edit, I chose Edit, and when I want to Run, I chose Run.  

 
As we make further modifications in the next sections, I will continue to Edit the Run_Monthly_Report.ctm 
and add SAS code to the CodeTemplate element. 

Note: After our custom task is finalized, it will be moved to a Production_SAS_Tasks\ folder on our production SAS 
Server and the users that run the monthly process will only have Read access, so while SAS may prompt them if they 
want to Edit the custom task, they will not have access to do it. This control is maintained with Active Directory 
Groups through our IT department so only a Deployment team can make changes to the production environment.  

NEW DIRECTORIES 
Each month we want to store all files related to our monthly report process. Typically, audit teams want to 
see all the data, programs, logs and files created when this process ran. Sometimes they will request we 
have the capability to rerun the entire process and show we get the same results. For this reason, it is 
important to save a snapshot of all SAS programs, SAS tasks, SAS logs, SAS datasets, and output from 
the monthly process. If any issues with the final monthly report are found, we can investigate because all 
the files related to the process are saved. We also can reproduce the exact results because we have the 
‘version’ of the SAS tasks, SAS datasets, and SAS programs saved that produced the results in the first 
place.  

This type of storage is easier done if we can save all the files to a folder with the month end date. It’s a lot 
less confusing than using the month end date in the name of the files and storing them all in one location. 
We don’t want to manually go add the new folder/directory; especially if we are running a process on a 
production server; we want SAS to do it for us.  

Now that we have custom task where the month end date is entered, we can easily add some SAS code 
to our CodeTemplate element that will create specific directories to store all files related to the monthly 
report process. 

The following code will create the directories we need. Note: it does not hurt to have this code run even if the 
directories or folders already exist. This particular logic will not delete and create new folders.*  The options 
dlcreatedir is very powerful; but be sure to turn it off after because it will literally create a folder with any 
libname statement, even if the folder does not exist! 

 
/* Create macro variable prodpath */ 
%let prodpath = xxxxxxxx\Production\Monthly_Reports; 
%put &prodpath.; 
 
/* Builds Folders for Monthly Report Process Artifacts */ 
%macro create_month_end_dir;   
  %if &end_month_dt. eq 1 %then %do;   
  options DLCREATEDIR; 
 LIBNAME FOLDER "&prodpath.\M_&dateMonthEnd."; 
 LIBNAME FOLDER "&prodpath.\M_&dateMonthEnd.\SAS_Data"; 
 LIBNAME FOLDER "&prodpath.\M_&dateMonthEnd.\SAS_Logs"; 
 LIBNAME FOLDER "&prodpath.\M_&dateMonthEnd.\Output"; 
 LIBNAME FOLDER "&prodpath.\M_&dateMonthEnd.\SAS_Programs"; 
   
    DATA QTR_END_msg; 
     format var $45.; 
     VAr= 'MONTH END FOLDERS CREATED'; 
   run; 
   title "Folder for Month End &dateMonthEnd were created"; 
   proc print;  
   run; 
   title; 
  %end; 
  %else %if &end_month_dt. eq 0 %then %do; 
   DATA error_msg; 
     format var $45.; 
     var= 'ERROR: MUST BE LAST DAY OF THE MONTH'; 
    run; 

    Title "Creating Folders for &dateMonthEnd. was aborted"; 
    proc print;  
    run; 
   title; 
  %end; 
%mend create_month_end_dir; 
%create_month_end_dir; 
 
/* Create macro variable mepath */ 
%let mepath = M_&dateMonthEnd.; 
%put &mepath.; 
 
options NODLCREATEDIR;; 

 

*It is possible to have SAS read the main directory and create a macro that determines if the folders we 
are asking SAS to create already exists. This seemed an additional layer of complexity, so we did not 
include it in this paper but please reach out to me if you are interested. It’s easily done with the use of the 
Filename pipedir statement, and Infile pipedir statement combined. 

As you can see in the code above, I like to create a folder for each type of file related to running a SAS 
process. I have found this allows users to find the files they need to see quickly, and I use it consistently 
with storage of all SAS processes. Also, I do not like starting a folder name with a number, so I always 
add a Character and I never have spaces in folder names. (This is probably left over from running SAS in 
a Unix environment). In my humble opinion, it’s very poor practice to allow spaces in names of folders, 
files, or variables!  

 Now that this code has been added we need to test and see that it works, but I already know of an issue 
that will cause an error. Remember how the custom task calls the SAS program 
Run_Monthly_Report.sas? 

%include "&prodpath.\&mepath.\SAS_Programs\Run_Monthly_Report.sas"; 
 

Since we now have SAS creating this directory/folder the SAS program will not be within it. We need SAS 
to copy it from what I like to call the Production_SAS_Program_Repository. This is a folder where the 
latest versions of the Production SAS programs are stored. So, we need to add another macro to our 
CodeTemplate element in order for SAS to copy Run_Monthly_Report.sas to our new folder location.  

%macro copy_prod_pgm; 
  %if &end_month_dt. eq 1 %then %do; 
    X "copy &prodpath.\Production_SAS_Programs_Repository\*.* 
&prodpath.\&mepath.\SAS_Programs\"; 
   
     DATA msg_prod_files; 
     format var $80.; 
     VAr= "COPIED Production SAS Program"; 
   run; 
   proc print;  
   run; 
  %end; 
  %else %if &end_month_dt. eq 0 %then %do; 
    DATA error_msg; 
     format var $45.; 
     var= 'ERROR: MUST BE LAST DAY OF THE MONTH'; 
    run; 
    Title "Copying SAS program for &dateMonthEnd. was aborted"; 
    proc print;  
    run; 
   title; 



20 

  %end;   
%mend copy_prod_pgm; 
%copy_prod_pgm; 
   
 

The two macros %create_month_end_dir and %copy_prod_pgm could probably be combined but I like to 
keep each action separate for transparency to users.  

Now we have added a macro to create month end folders and a macro to copy the SAS program so the 
%include statement to call Run_Monthly_Report.sas will not cause an error. The screenshot below shows 
the results of running the updated Run_Monthly_Report.ctm.  

 
 

The custom task ran and produced the proper messages as we designed. The folders were created and 
the Run_Monthly_Report.sas program was copied. The report gets generated since the code in the 
Run_Monthly_Report.sas program is executed. 

These screenshots below show the folder was created, and the Run_Monthly_Report.sas program was 
copied to the M_&dateMonthEnd\SAS_Programs\ location.  

The output was also saved to the location specified in the Run_Monthly_Report.sas program. 

  
By adding the additional SAS macro language we have automated the production of the folders; as well 
as saved a version of the actual SAS program that ran, and the output that was generated.  

VERSION CONTROL 
There are many ways to control versions of SAS programs. For our purposes we are going to 
demonstrate one of the simplest.  

If we modify our folder structure and naming conventions, we can make use of some version control. At 
this point we can modify the custom task name to Run_Monthly_Report_v1.ctm.  

Next, if we modify the folder structure to add a Version_1\ subfolder to 
Production_SAS_Programs_Repository we can also control which version of the SAS program gets 
copied and executed. This new folder structure is shown in the screenshot below. 

 
If you recall the code, we added to the macro %copy_prod_pgm; it was written to copy all files from 
Production_SAS_Programs_Repository\ to our new month end SAS_Programs\ folder.  

 
X "copy &prodpath.\Production_SAS_Programs_Repository\*.* 
&prodpath.\&mepath.\SAS_Programs\"; 
 

Now we want it to copy all files from the latest version subfolder. There are a few different ways this can 
be done programmatically but to keep it simple we are just going to add a new input option to our custom 
task asking the user to identify the Version of SAS programs that should be used. This will require the 
following new code to get added to both the Metadata-Options and UI elements, as well as some 
additional SAS code to the CodeTemplate element. 

This is added to the Metadata-Options element: 

 
Note: the text option is set with a default value of 1. We did not implement any controls to verify that the version 
number entered exists. If a number or character is entered that does not exist, it will cause an error. SAS macro 
language can be added to check the number entered actually exists in a folder name with 
Production_SAS_Programs_Repository\ but we will not show it in this paper.   

 

This is added to the UI element: 

This is added to the CodeTemplate element: 

 
 

The following SAS code was modified in the %macro copy_prod_pgms so the &VersionNum. macro 
variable is used in the path name and the message to the user.  

 
The RESULTS tab now looks like the screenshot below when the Run_Monthly_Report.ctm is ran. 

 
 

 

 

The option and code additions allow some version control. With these changes new versions can be 
deployed without the production user having to touch any SAS programs.  

If a new active directory group is created that only allows a deployment team to have write access to the 
Production_SAS_Programs_Repository\ folder then they can follow IT change management processes to 
deploy new versions.  

We still have not added a means to automatically save the SAS log files after running the custom task. 
This is the next step in automating the monthly report process.  

LOG STORAGE AND CHECKING 
Checking the SAS log files after running SAS programs is very important. Many times users can miss 
errors and may send out results before realizing an error occurred in the process.   

The PROC PRINTTO statement routes SAS log and procedure output directly to designated file and 
location. This will get added so all the SAS code is captured and stored.  

This time we will make the change to the Run_Monthly_Report SAS program. We will add the PROC 
PRINTTO statement as shown below in the boxes at the top of the program and at the bottom. 

 
I also added code so the SAS log file saved will also show the values of the macro variables defined in 
the custom task.  

Now when we look at the LOG tab after the custom task runs it will no longer show the code from running 
Run_Monthly_Reports.sas, instead the PROCEDURE PRINTTO used is listed. 

 The SAS log file for Run_Monthly_Report.sas has been saved and stored in the location as specified.  

 
The disadvantage to writing the SAS log file to a specified file is if there was an Error in executing the 
code in Run_Monthly_Report.sas it will not show in the LOG tab, or even in the Errors, Warnings, Notes 
section of the LOG tab.  

 
 

Luckily there have been many papers offering SAS macro language that reads the saved SAS log file and 
provide a report if errors are encountered. We copied the %checklogs macro from a paper by Richann 
Watson called “Check Please: An Automated Approach to Log Checking”.   

With a few modifications to the %checklogs macro from this paper, it is used to read our saved SAS log 
file and produce a report. The macro is added to the custom task along with the following statement to 
call the macro. 

%checklogs(loc=&prodpath.\&mepath.\SAS_Logs, out=Run_Monthly_Report); 
 

After running the custom task with this additional functionality, the RESULTS tab shows a Report of all 
WARNINGS, ERRORS or NOTES.  

It also saves the Report as a pdf file to the SAS_Logs\ folder location, see the screenshot below.  

 
The Run_Monthly_Report15AUG220132.PDF shows the following information. This means not only is the 
SAS log file saved after running the monthly process, but a report showing it occurred without errors is 
also produced and stored for any audit reviews.  

 
What happens if there is an error in the SAS log file? I’ve made a slight change to the SAS code to 
generate an error and the screenshot below shows the RESULTS tab if an error is found by the 
%checklogs macro programming.  

 
At this point we have fully automated running a monthly report process in SAS Studio. When the custom 
task runs, the SAS log file, SAS program, and output reports are all saved in specific folders built by the 
custom task. The saved SAS log file is reviewed for errors and warnings, and all of it is saved for an audit 

review later. The user running the monthly report process has not had to touch any folder, file or program 
in order to produce the monthly report. SAS has done all the work! 

MORE AUTOMATION TECHNIQUES – EMAIL REPORTS 
While we have shown many techniques to automate a monthly process for production there are still a few 
more techniques that can be helpful. 

EMAIL REPORTS 
The report produced by the custom task Run_Monthly_Report.ctm is saved to an Output\ folder as 
SOX_SQL_Snapshot_MonthEnd.html. How is this report viewed by stakeholders? Even if we have the 
capability to publish the report to a SharePoint or web site, sometimes audit needs proof that the report 
was received by an executive stakeholder. This can be done by attaching the report and emailing it to the 
person.  

The FILENAME EMAIL statement allows SAS to send email. It also allows to specify all component parts 
of an email message, including Subject, Recipients (To, CC, BCC), From and more.  

We will write a new program called Email_Monthly_Report.sas that sends an email to our executive 
stakeholder with the report attached. We will also modify the custom task with some additional logic to run 
this new SAS program.  

The screenshot below shows the contents of this new SAS program. Notice we still want to capture a 
SAS log file when it executes to ensure it also doesn’t encounter any errors.  

The screenshot below shows a new macro %email_monthly_report added to the custom task in the 
CodeTemplate element. It will run the Email_Monthly_Report.sas program if the SAS log file from the 
execution of Run_Monthly_Report.sas did not have any errors.  

The macro variable &logcnt. was created when the %checklogs macro executed for Run_Monthly_Report 
and will be greater than 0 if any errors are found. We also included the same conditions used previously 
in the macro %run_monthly_report requiring the &chkData. eq 1 and the &end_month_dt. eq 1. This new 
code will also produce a message if Email_Monthly_Report.sas is not ran due to an error in the SAS log 
from Run_Monthly_Report.sas.   

 
As you can see, we are utilizing the same concept as done previously, by using %include statement to 
call the new SAS program, and using %checklogs to check the SAS log for Email_Monthly_Report. 

When the custom task runs with this additional SAS macro language the RESULTS are shown in the 
screenshot below.  

We can also see the new SAS log file saved in the directory.  

 
Finally, the screenshot below shows the email that is sent by SAS to our stakeholder.  

 
 Now that SAS is also sending emails with the monthly report attached, nearly every aspect of the 
process is automated-- all with the click of a button from our custom task! 

I haven’t expanded on the use of the &chkData. macro variable created in the custom task, but as you 
can imagine this could be used for other conditional logic in our SAS program if desired. For this example, 
it simply acknowledges receipt of some type was received by the production user running the custom task 
indicating the data stored in SQL was updated for the month end and ready to be processed.  

CONCLUSION 
In conclusion, SAS Studio offers a powerful tool for automation with a custom task. By building your own 
point-and-click user interface an entire manual process can be automated. As a SAS programmer I took 
the lazy route and used the code found in the Sample Task to build my custom task. This shows it can be 
done without learning an entire new language! If you already have your SAS Server integrated with your 
IT department under full production controls, then the techniques shown in this paper may not be as 
useful to you; but for those who are running consistent processes without any controls I hope you are 
able to see how a custom task can help with your next audit review. Even if you are managing processes 
that are not reviewed by audit these techniques can still reduce time and errors by reducing any manual 
processing that may otherwise be required. I have always said “Let SAS do the work!” and after twenty 
years working with SAS products I stand by my mantra. SAS has the capability to meet the most stringent 
audit requirements with just Base SAS, SAS macro language and a custom task. I hope you found this 
paper helpful, and it taught you some new tricks. See you at the next conference! Keep learning SAS!  

